;; -*- coding: utf-8 -*-
(eval-when (:compile-toplevel :load-toplevel :execute)
  (defparameter OPT
    #+swank '(optimize (speed 3) (safety 2))
    #-swank '(optimize (speed 3) (safety 0) (debug 0)))
  #+swank (progn (ql:quickload '(:cl-debug-print :fiveam))
                 (shadow :run)
                 (use-package :fiveam)))
#+swank (cl-syntax:use-syntax cl-debug-print:debug-print-syntax)

;; BEGIN_INSERTED_CONTENTS
(defmacro define-binary-heap (name &key (order '#'>) (element-type 'fixnum))
  (check-type name symbol)
  (let* ((string-name (string name))
         (fname-push (intern (format nil "~A-PUSH" string-name)))
         (fname-pop (intern (format nil "~A-POP" string-name)))
         (fname-reinitialize (intern (format nil "~A-REINITIALIZE" string-name)))
         (fname-empty-p (intern (format nil "~A-EMPTY-P" string-name)))
         (fname-peak (intern (format nil "~A-PEAK" string-name)))
         (fname-make (intern (format nil "MAKE-~A" string-name)))
         (acc-position (intern (format nil "~A-POSITION" string-name)))
         (acc-data (intern (format nil "~A-DATA" string-name))))
    `(progn
       (defstruct (,name
                   (:constructor ,fname-make
                       (size
                        &aux (data ,(if (eql element-type '*)
                                        `(make-array (1+ size))
                                        `(make-array (1+ size) :element-type ',element-type))))))
         (data #() :type (simple-array ,element-type (*)) :read-only t)
         (position 1 :type (integer 1 #.most-positive-fixnum)))

       (declaim (inline ,fname-push))
       (defun ,fname-push (obj heap)
         (symbol-macrolet ((position (,acc-position heap)))
           (let ((data (,acc-data heap)))
             (declare ((simple-array ,element-type (*)) data))
             (labels ((update (pos)
                        (unless (= pos 1)
                          (let ((parent-pos (ash pos -1)))
                            (when (funcall ,order (aref data pos) (aref data parent-pos))
                              (rotatef (aref data pos) (aref data parent-pos))
                              (update parent-pos))))))
               (setf (aref data position) obj)
               (update position)
               (incf position)
               heap))))

       (declaim (inline ,fname-pop))
       (defun ,fname-pop (heap &optional (error t) null-value)
         (symbol-macrolet ((position (,acc-position heap)))
           (let ((data (,acc-data heap)))
             (declare ((simple-array ,element-type (*)) data))
             (labels ((update (pos)
                        (declare ((integer 1 #.most-positive-fixnum) pos))
                        (let* ((child-pos1 (+ pos pos))
                               (child-pos2 (1+ child-pos1)))
                          (when (<= child-pos1 position)
                            (if (<= child-pos2 position)
                                (if (funcall ,order (aref data child-pos1) (aref data child-pos2))
                                    (unless (funcall ,order (aref data pos) (aref data child-pos1))
                                      (rotatef (aref data pos) (aref data child-pos1))
                                      (update child-pos1))
                                    (unless (funcall ,order (aref data pos) (aref data child-pos2))
                                      (rotatef (aref data pos) (aref data child-pos2))
                                      (update child-pos2)))
                                (unless (funcall ,order (aref data pos) (aref data child-pos1))
                                  (rotatef (aref data pos) (aref data child-pos1))))))))
               (if (= position 1)
                   (if error
                       (error "No element in heap.")
                       null-value)
                   (prog1 (aref data 1)
                     (decf position)
                     (setf (aref data 1) (aref data position))
                     (update 1)))))))

       (declaim (inline ,fname-reinitialize))
       (defun ,fname-reinitialize (heap)
         (setf (,acc-position heap) 1)
         heap)

       (declaim (inline ,fname-empty-p))
       (defun ,fname-empty-p (heap)
         (= 1 (,acc-position heap)))

       (declaim (inline ,fname-peak))
       (defun ,fname-peak (heap &optional (error t) null-value)
         (if (= 1 (,acc-position heap))
             (if error
                 (error "No element in heap")
                 null-value)
             (aref (,acc-data heap) 1))))))

(declaim (ftype (function * (values fixnum &optional)) read-fixnum))
(defun read-fixnum (&optional (in *standard-input*))
  (declare #.OPT)
  (macrolet ((%read-byte ()
               `(the (unsigned-byte 8)
                     #+swank (char-code (read-char in nil #\Nul))
                     #-swank (sb-impl::ansi-stream-read-byte in nil #.(char-code #\Nul) nil))))
    (let* ((minus nil)
           (result (loop (let ((byte (%read-byte)))
                           (cond ((<= 48 byte 57)
                                  (return (- byte 48)))
                                 ((zerop byte) ; #\Nul
                                  ;; (return-from read-fixnum 0)
                                  (error "Read EOF or #\Nul."))
                                 ((= byte #.(char-code #\-))
                                  (setf minus t)))))))
      (declare ((integer 0 #.most-positive-fixnum) result))
      (loop
        (let* ((byte (%read-byte)))
          (if (<= 48 byte 57)
              (setq result (+ (- byte 48) (* 10 (the (integer 0 #.(floor most-positive-fixnum 10)) result))))
              (return (if minus (- result) result))))))))

(defmacro dbg (&rest forms)
  #+swank
  (if (= (length forms) 1)
      `(format *error-output* "~A => ~A~%" ',(car forms) ,(car forms))
      `(format *error-output* "~A => ~A~%" ',forms `(,,@forms)))
  #-swank (declare (ignore forms)))

(defmacro define-int-types (&rest bits)
  `(progn
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "UINT~A" b)) () '(unsigned-byte ,b))) bits)
     ,@(mapcar (lambda (b) `(deftype ,(intern (format nil "INT~A" b)) () '(signed-byte ,b))) bits)))
(define-int-types 2 4 7 8 15 16 31 32 62 63 64)

(declaim (inline println))
(defun println (obj &optional (stream *standard-output*))
  (let ((*read-default-float-format* 'double-float))
    (prog1 (princ obj stream) (terpri stream))))

(defconstant +mod+ 1000000007)

;; Body

(define-binary-heap heap
  :order (lambda (x y) (> (car x) (car y)))
  ;; (cal . index)
  :element-type (cons fixnum fixnum))

(defun solve (xs vs c)
  (declare #.OPT
           ((simple-array uint62 (*)) xs)
           ((simple-array uint32 (*)) vs))
  (let* ((n (length xs))
         (cumuls-clock (make-array n :element-type 'fixnum))
         (cumuls-anticlock (make-array n :element-type 'fixnum)))
    (setf (aref cumuls-clock 0) (- (aref vs 0) (aref xs 0)))
    (loop for i from 1 below n
          do (setf (aref cumuls-clock i)
                   (+ (aref cumuls-clock (- i 1))
                      (aref vs i)
                      (- (aref xs (- i 1)) (aref xs i)))))
    (setf (aref cumuls-anticlock (- n 1)) (- (aref vs (- n 1))
                                             (* 2 (- c (aref xs (- n 1))))))
    (loop for i from (- n 2) downto 0
          do (setf (aref cumuls-anticlock i)
                   (+ (aref cumuls-anticlock (+ i 1))
                      (aref vs i)
                      (- (* 2 (- (aref xs (+ i 1)) (aref xs i)))))))
    (let ((res (reduce #'max cumuls-clock))
          (pqueue (make-heap n))
          (pos 0))
      (declare (fixnum res))
      (dotimes (i n)
        (heap-push (cons (aref cumuls-anticlock i) i) pqueue))
      (loop (when (heap-empty-p pqueue)
              (return res))
            (destructuring-bind (rev-cal . rev-pos) (heap-peak pqueue)
              (declare (fixnum rev-cal rev-pos))
              (if (<= rev-pos pos)
                  (heap-pop pqueue)
                  (progn
                    (setf res (max res (+ (aref cumuls-clock pos) rev-cal)))
                    (incf pos))))))))

(defun main ()
  (declare #.OPT)
  (let* ((n (read))
         (c (read))
         (xs (make-array n :element-type 'uint62))
         (rev-xs (make-array n :element-type 'uint62))
         (vs (make-array n :element-type 'uint32))
         (rev-vs (make-array n :element-type 'uint32)))
    (declare (uint62 c) (uint32 n))
    (dotimes (i n)
      (let ((x (read-fixnum))
            (v (read-fixnum)))
        (setf (aref xs i) x)
        (setf (aref rev-xs (- n i 1)) (- c x))
        (setf (aref vs i) v)
        (setf (aref rev-vs (- n i 1)) v)))
    (println (max 0 (solve xs vs c) (solve rev-xs rev-vs c)))))

#-swank(main)


Measured memory_bytes: 41960